1 Példa 1:

Adatok megadása:

```
(%i1)   kill(all);
(%o0)   done
```

```
(%i10)  A1:60$ /* mm^2-ben */
        A2:20$ /* mm^2-ben */
        A3:30$ /* mm^2-ben */
        E1:100e3$ /* MPa-ban */
        E2:200e3$ /* MPa-ban */
        E3:50e3$ /* MPa-ban */
        L1:1e3$ /* mm-ben */
        L2:2e3$ /* mm-ben */
        L3:3e3$ /* mm-ben */
        Ft:15e3$ /* N-ban */;
```

Globális csomóponti elmozdulásvektor és terhelésvektor:

```
(%i12)  U:matrix([u1],[u2],[u3]);
        F:matrix([F1],[F2],[F3]);
```

(U) $\begin{bmatrix} u1 \\ u2 \\ u3 \end{bmatrix}$

(F) $\begin{bmatrix} F1 \\ F2 \\ F3 \end{bmatrix}$

Elem-csomópont összerendelések tárolása például az en mátrixban:

```
(%i13)  en:matrix([1,2],[2,3],[2,3]);
```

(en) $\begin{bmatrix} 1 & 2 \\ 2 & 3 \\ 2 & 3 \end{bmatrix}$

Elemek merevségei:

```
(%i17)  k(a,e,l):=a*e/l$
        k1:k(A1,E1,L1);
        k2:k(A2,E2,L2);
        k3:k(A3,E3,L3);
(k1)    6000.0
(k2)    2000.0
(k3)    500.0
```

Elem merevségi mátrixok:

```
(%i20) K1:k1*matrix([1,-1],[-1,1]);
       K2:k2*matrix([1,-1],[-1,1]);
       K3:k3*matrix([1,-1],[-1,1]);
```

(K1) $\begin{bmatrix} 6000.0 & -6000.0 \\ -6000.0 & 6000.0 \end{bmatrix}$

(K2) $\begin{bmatrix} 2000.0 & -2000.0 \\ -2000.0 & 2000.0 \end{bmatrix}$

(K3) $\begin{bmatrix} 500.0 & -500.0 \\ -500.0 & 500.0 \end{bmatrix}$

Globális merevségi mátrix megadása során első lépésben egy zérus elemekkel kitöltött mátrixot hozunk létre, majd a megfelelő helyekre betesszül az egyes elemem merevségi mátrixainak elemeit.
Egy lehetséges leprogramozása ennek az alábbiakban látható, ahol felhasználjuk az elem-csomópont összerendlés mátrixot:

```
(%i21) KG:matrix( [0,0,0],  [0,0,0],  [0,0,0])$
```

Az 1-es elem merevségi mátrixának elhelyezése a globális merevségi mátrixban manuálisan:

```
(%i25) KG[en[1,1],en[1,1]]:K1[1,1]$
       KG[en[1,1],en[1,2]]:K1[1,2]$
       KG[en[1,2],en[1,1]]:K1[2,1]$
       KG[en[1,2],en[1,2]]:K1[2,2]$
```

Az 1-es elem merevségi mátrixának elhelyezése a globális merevségi mátrixban ciklus segítségével:

```
(%i26) KG:matrix( [0,0,0],  [0,0,0],  [0,0,0])$
```

```
(%i27) for iii: 1 step 1 thru 2 do
           for jjj : 1 step 1 thru 2 do
               KG[en[1,iii],en[1,jjj]]:KG[en[1,iii],en[1,jjj]]+K1[iii,jjj]$
```

A 2-es elem merevségi mátrixának elhelyezése a globális merevségi mátrixban:

```
(%i28) for iii: 1 step 1 thru 2 do
           for jjj : 1 step 1 thru 2 do
               KG[en[2,iii],en[2,jjj]]:KG[en[2,iii],en[2,jjj]]+K2[iii,jjj]$
```

A 3-as elem merevségi mátrixának elhelyezése a globális merevségi mátrixban:

```
(%i29) for iii: 1 step 1 thru 2 do
           for jjj : 1 step 1 thru 2 do
               KG[en[3,iii],en[3,jjj]]:KG[en[3,iii],en[3,jjj]]+K3[iii,jjj]$
```

Tehát a globális merevségi mátrix:

```
(%i30) KG;
```

(%o30) $\begin{bmatrix} 6000.0 & -6000.0 & 0 \\ -6000.0 & 8500.0 & -2500.0 \\ 0 & -2500.0 & 2500.0 \end{bmatrix}$

Csomóponti terhelések megadása:

```
(%i31) F:F,F1=Ft,F2=0;
```

(F) $\begin{bmatrix} 15000.0 \\ 0 \\ F3 \end{bmatrix}$

Peremfeltétel figyelembe vétele:

```
(%i32) U:U,u3=0;
```

(U) $\begin{bmatrix} u1 \\ u2 \\ 0 \end{bmatrix}$

Kondenzált merevségi mátrixot megkapjuk a 3 sorok/oszlopok törlésével:

```
(%i33) KGkond:submatrix(3,KG,3);
```

(KGkond) $\begin{bmatrix} 6000.0 & -6000.0 \\ -6000.0 & 8500.0 \end{bmatrix}$

A kondenzált tehervektort megkapjuk a 3. elem törlésével:

```
(%i34) Fkond:submatrix(3,F);
```

(Fkond) $\begin{bmatrix} 15000.0 \\ 0 \end{bmatrix}$

Megoldás az elmozdulásokra:

```
(%i35) mego:first(linsolve_by_lu(KGkond,Fkond));
0 errors, 0 warnings
rat: replaced 6000.0 by 6000/1 = 6000.0
```

(mego) $\begin{bmatrix} 8.5 \\ 6.0 \end{bmatrix}$

Tehát a globális csomóponti elmozdulásvektor:

```
(%i36) U:U,u1=mego[1,1],u2=mego[2,1];
```

$$(U)\quad \begin{bmatrix} 8.5 \\ 6.0 \\ 0 \end{bmatrix}$$

Csomóponti terhelések vektora:

```
(%i37) KG.U;
```

$$(\%o37)\quad \begin{bmatrix} 15000.0 \\ 0.0 \\ -15000.0 \end{bmatrix}$$

Az egyes elmekhez tartozó lokális elmozdulásvektorok:

```
(%i40) Ue1:matrix(U[en[1,1]],U[en[1,2]]);
Ue2:matrix(U[en[2,1]],U[en[2,2]]);
Ue3:matrix(U[en[3,1]],U[en[3,2]]);
```

$$(Ue1)\quad \begin{bmatrix} 8.5 \\ 6.0 \end{bmatrix}$$

$$(Ue2)\quad \begin{bmatrix} 6.0 \\ 0 \end{bmatrix}$$

$$(Ue3)\quad \begin{bmatrix} 6.0 \\ 0 \end{bmatrix}$$

Az egyes elmekhez tartozó lokális tehervektorok:

```
(%i43) Fe1:K1.Ue1;
Fe2:K2.Ue2;
Fe3:K3.Ue3;
```

$$(Fe1)\quad \begin{bmatrix} 15000.0 \\ -15000.0 \end{bmatrix}$$

$$(Fe2)\quad \begin{bmatrix} 12000.0 \\ -12000.0 \end{bmatrix}$$

$$(Fe3)\quad \begin{bmatrix} 3000.0 \\ -3000.0 \end{bmatrix}$$

Tehát a normál igénybevételek:

```
(%i46) N1:Fe1[2][1];
N2:Fe2[2][1];
N3:Fe3[2][1];
```

(N1) -15000.0
(N2) -12000.0
(N3) -3000.0

Vagyis mindegyik rúd nyomó igénybevétel alatt van.

A rudakban ébredő feszültségek:

```
(%i49)  σ1:N1/A1;
        σ2:N2/A2;
        σ3:N3/A3;
(σ1)    -250.0
(σ2)    -600.0
(σ3)    -100.0
```

A rudak alakváltozásai:

```
(%i52)  ε1:σ1/E1;
        ε2:σ2/E2;
        ε3:σ3/E3;
(ε1)    -0.0025
(ε2)    -0.003
(ε3)    -0.002
```

2 Példa 2:

Adatok megadása:

```
(%i53)  kill(all);
(%o0)   done
```

```
(%i6)   A1:50$ /* mm^2-ben */
        A2:20$ /* mm^2-ben */
        E1:100e3$ /* MPa-ban */
        E2:200e3$ /* MPa-ban */
        L:2e3$ /* mm-ben */
        Ft:60e3$ /* N-ban */;
```

Globális csomóponti elmozdulásvektor és terhelésvektor:

```
(%i8)   U:matrix([u1],[u2],[u3],[u4])$
        F:matrix([F1],[F2],[F3],[F4])$
```

Elem-csomópont összerendelések tárolása például az en mátrixban:

```
(%i9)   en:matrix([1,2],[2,3],[4,2],[2,3]);
```

(en) $\begin{bmatrix} 1 & 2 \\ 2 & 3 \\ 4 & 2 \\ 2 & 3 \end{bmatrix}$

Elemek merevségei:

```
(%i14)  k(a,e,l):=a*e/l$
        k1:k(A1,E1,L);
        k2:k(A1,E1,L);
        k3:k(A1,E1,L);
        k4:k(A2,E2,2*L);
(k1)    2500.0
(k2)    2500.0
(k3)    2500.0
(k4)    1000.0
```

Elem merevségi mátrixok:

```
(%i18)  K1:k1*matrix([1,-1],[-1,1]);
        K2:k2*matrix([1,-1],[-1,1]);
        K3:k3*matrix([1,-1],[-1,1]);
        K4:k4*matrix([1,-1],[-1,1]);
```

(K1) $\begin{bmatrix} 2500.0 & -2500.0 \\ -2500.0 & 2500.0 \end{bmatrix}$

(K2) $\begin{bmatrix} 2500.0 & -2500.0 \\ -2500.0 & 2500.0 \end{bmatrix}$

(K3) $\begin{bmatrix} 2500.0 & -2500.0 \\ -2500.0 & 2500.0 \end{bmatrix}$

(K4) $\begin{bmatrix} 1000.0 & -1000.0 \\ -1000.0 & 1000.0 \end{bmatrix}$

Globális merevségi mátrix megadása során első lépésben egy zérus elemekkel kitöltött mátrixot hozunk létre, majd a megfelelő helyekre betesszük az egyes elemem merevségi mátrixainak elemeit.
Egy lehetséges leprogramozása ennek az alábbiakban látható, ahol felhasználjuk az elem-csomópont összerendlés mátrixot:

```
(%i19)  KG: matrix(
         [0,0,0,0],
         [0,0,0,0],
         [0,0,0,0],
         [0,0,0,0]
        )$
```

Az 1-es elem merevségi mátrixának elhelyezése a globális merevségi mátrixban:

```
(%i20)  for iii: 1 step 1 thru 2 do
            for jjj : 1 step 1 thru 2 do
                KG[en[1,iii],en[1,jjj]]:KG[en[1,iii],en[1,jjj]]+K1[iii,jjj]$
```

A 2-es elem merevségi mátrixának elhelyezése a globális merevségi mátrixban:

```
(%i21)  for iii: 1 step 1 thru 2 do
            for jjj : 1 step 1 thru 2 do
                KG[en[2,iii],en[2,jjj]]:KG[en[2,iii],en[2,jjj]]+K2[iii,jjj]$
```

A 3-as elem merevségi mátrixának elhelyezése a globális merevségi mátrixban:

```
(%i22)  for iii: 1 step 1 thru 2 do
            for jjj : 1 step 1 thru 2 do
                KG[en[3,iii],en[3,jjj]]:KG[en[3,iii],en[3,jjj]]+K3[iii,jjj]$
```

A 4-es elem merevségi mátrixának elhelyezése a globális merevségi mátrixban:

```
(%i23)  for iii: 1 step 1 thru 2 do
            for jjj : 1 step 1 thru 2 do
                KG[en[4,iii],en[4,jjj]]:KG[en[4,iii],en[4,jjj]]+K4[iii,jjj]$
```

Tehát a globális merevségi mátrix:

```
(%i24)  KG;
```

(%o24) $$\begin{bmatrix} 2500.0 & -2500.0 & 0 & 0 \\ -2500.0 & 8500.0 & -3500.0 & -2500.0 \\ 0 & -3500.0 & 3500.0 & 0 \\ 0 & -2500.0 & 0 & 2500.0 \end{bmatrix}$$

Csomóponti terhelések megadása:

```
(%i25)  F:F,F4=-Ft,F2=0;
```

(F) $$\begin{bmatrix} F1 \\ 0 \\ F3 \\ -60000.0 \end{bmatrix}$$

Peremfeltétel figyelembe vétele:

```
(%i26)  U:U,u1=0,u3=0;
```

(U) $$\begin{bmatrix} 0 \\ u2 \\ 0 \\ u4 \end{bmatrix}$$

Kondenzált merevségi mátrixot megkapjuk az 1,3 sorok/oszlopok törlésével:

```
(%i27) KGkond:submatrix(1,3,KG,1,3);
```

$$(KGkond)\quad \begin{bmatrix} 8500.0 & -2500.0 \\ -2500.0 & 2500.0 \end{bmatrix}$$

A kondenzált tehervektort megkapjuk az 1,3 sorok törlésével:

```
(%i28) Fkond:submatrix(1,3,F);
```

$$(Fkond)\quad \begin{bmatrix} 0 \\ -60000.0 \end{bmatrix}$$

Megoldás az elmozdulásokra:

```
(%i29) mego:first(linsolve_by_lu(KGkond,Fkond));
rat: replaced 8500.0 by 8500/1 = 8500.0
```

$$(mego)\quad \begin{bmatrix} -10.000000000000002 \\ -34.00000000000001 \end{bmatrix}$$

Tehát a globális csomóponti elmozdulásvektor:

```
(%i30) U:U,u2=mego[1,1],u4=mego[2,1];
```

$$(U)\quad \begin{bmatrix} 0 \\ -10.000000000000002 \\ 0 \\ -34.00000000000001 \end{bmatrix}$$

Csomóponti terhelések vektora:

```
(%i31) KG.U;
```

$$(\%o31)\quad \begin{bmatrix} 25000.0 \\ 0.0 \\ 35000.00000000001 \\ -60000.00000000001 \end{bmatrix}$$

Az egyes elmekhez tartozó lokális elmozdulásvektorok:

```
(%i35) Ue1:matrix(U[en[1,1]],U[en[1,2]]);
       Ue2:matrix(U[en[2,1]],U[en[2,2]]);
       Ue3:matrix(U[en[3,1]],U[en[3,2]]);
       Ue4:matrix(U[en[4,1]],U[en[4,2]]);
```

(Ue1) $\begin{bmatrix} 0 \\ -10.000000000000002 \end{bmatrix}$

(Ue2) $\begin{bmatrix} -10.000000000000002 \\ 0 \end{bmatrix}$

(Ue3) $\begin{bmatrix} -34.00000000000001 \\ -10.000000000000002 \end{bmatrix}$

(Ue4) $\begin{bmatrix} -10.000000000000002 \\ 0 \end{bmatrix}$

Az egyes elmekhez tartozó lokális tehervektorok:

```
(%i39) Fe1:K1.Ue1;
       Fe2:K2.Ue2;
       Fe3:K3.Ue3;
       Fe4:K4.Ue4;
```

(Fe1) $\begin{bmatrix} 25000.0 \\ -25000.0 \end{bmatrix}$

(Fe2) $\begin{bmatrix} -25000.0 \\ 25000.0 \end{bmatrix}$

(Fe3) $\begin{bmatrix} -60000.00000000001 \\ 60000.00000000001 \end{bmatrix}$

(Fe4) $\begin{bmatrix} -10000.000000000002 \\ 10000.000000000002 \end{bmatrix}$

Tehát a normál igénybevételek:

```
(%i43) N1:Fe1[2][1];
       N2:Fe2[2][1];
       N3:Fe3[2][1];
       N4:Fe4[2][1];
```

(N1) -25000.0
(N2) 25000.0
(N3) 60000.00000000001
(N4) 10000.000000000002

A rudakban ébredő feszültségek:

```
(%i47)  σ1:N1/A1;
        σ2:N2/A1;
        σ3:N3/A1;
        σ4:N4/A2;
(σ1)    -500.0000000000001
(σ2)    500.0000000000001
(σ3)    1200.0
(σ4)    500.0000000000001
```

A rudak alakváltozásai:

```
(%i51)  ε1:σ1/E1;
        ε2:σ2/E1;
        ε3:σ3/E1;
        ε4:σ4/E2;
(ε1)    -0.005000000000000001
(ε2)    0.005000000000000001
(ε3)    0.012
(ε4)    0.0025
```