Minden adat törlése, Temp könyvtár beállítása:

(%i1)
```
kill(all)$
maxima_tempdir:"c:/temp"$
```

Kiindulási adatok megadása:

(%i2)
```
adat:[L=3.0,F=7000.0, IE=200000.0]$
```

1 Reakcióerők számítása

A lehetséges reakcióerők a megtámasztásokból adódóan: Ax, Ay, By
A 3 statikai egyenlet (két erő-egyensúlyi egyenlet és 1 nyomatéki egyenlet például az A pontra felírva):

(%i5)
```
stat1:Ax=0$
stat2:Ay+By−F=0$
stat3:By·L/2−F·L=0$
```

Statikai egyenletek megoldása:

(%i6)
```
reak:linsolve([stat1,stat2,stat3],[Ax,Ay,By]);
```

(reak) $[Ax=0, Ay=-F, By=2\,F]$

Most már felírhatóak a nyomatéki függvények mindkét szakaszon:

(%i8)
```
Mh1:−Ay·x,reak;
Mh2:−Ay·x−By·(x−L/2),reak,ratsimp;
```

(Mh1) $F\,x$

(Mh2) $F\,L-F\,x$

2 Rugalmas szál differenciálegyenlete

A rugalmas szál diffegyenlete a két szakaszon: y1''(x)=-Mh1(x)/IE és y2''(x)=-Mh2(x)/IE. Itt most y1(x) és y2(x) függvények jelölik a lehajlásokat.
A lehajlásfüggvények számításához kétszer kell integrálni nem elfeledkezve az integrálási konstansokról! Legyen fi1(x)=y1'(x) függvénnyel jelölve a szögelfordulás függvény.

(%i10)
```
fi1:integrate(−Mh1/IE,x)+c1;
y1:integrate(fi1,x)+c2;
```

(fi1) $c1-\frac{F\,x^2}{2\,IE}$

(y1) $-\frac{F\,x^3}{6\,IE}+c1\,x+c2$

(%i12)
```
fi2:integrate(−Mh2/IE,x)+c3;
y2:integrate(fi2,x)+c4;
```

(fi2) $\frac{\frac{F\,x^2}{2}-F\,L\,x}{IE}+c3$

(y2) $\frac{\frac{F\,x^3}{6}-\frac{F\,L\,x^2}{2}}{IE}+c3\,x+c4$

Az ismeretlen c1,c2,c3,c4 kiszámításához szükség van a peremfeltételek megadására. Jelen esetben van két peremfeltételünk, miszerint az A és B helyeken a lehajlás zérus. Valamint van két illesztési feltételünk, hogy az y1 és y2 függvények értékei és első deriváltjai azonosak a B helyen.

(%i16)
```
pf1:ev(y1,x=0)=0$
pf2:ev(y1,x=L/2)=0$
pf3:ev(y1−y2,x=L/2)=0$
pf4:ev(fi1−fi2,x=L/2)=0$
```

Megoldás számítása:

(%i17)
```
cmego:linsolve([pf1,pf2,pf3,pf4],[c1,c2,c3,c4]);
```

(cmego) $[c1=\frac{F\,L^2}{24\,IE}, c2=0, c3=\frac{7\,F\,L^2}{24\,IE}, c4=-\frac{F\,L^3}{24\,IE}]$

Tehát a lehajlás és szögelfordulás függvények paraméteresen és numerikusan:

(%i21)
```
y1,cmego,ratsimp;
y1num:%,adat,expand;
y2,cmego,ratsimp;
y2num:%,adat,expand;
```

(%o18) $-\frac{4\,F\,x^3-F\,L^2\,x}{24\,IE}$

(y1num) $0.013125\,x-0.005833333333333334\,x^3$

(%o20) $\frac{4\,F\,x^3-12\,F\,L\,x^2+7\,F\,L^2\,x-F\,L^3}{24\,IE}$

(y2num) $0.005833333333333334\,x^3-0.0525\,x^2+0.091875\,x-0.039375$

(%i25)
```
fi1,cmego,ratsimp;
fi1num:%,adat,expand;
fi2,cmego,ratsimp;
fi2num:%,adat,expand;
```

(%o22) $-\frac{12\,F\,x^2-F\,L^2}{24\,IE}$

(fi1num) $0.013125-0.0175\,x^2$

(%o24) $\frac{12\,F\,x^2-24\,F\,L\,x+7\,F\,L^2}{24\,IE}$

(fi2num) $0.0175\,x^2-0.105\,x+0.091875$

3 *Ábrázolás*

Ábrázoljuk mindkét lehajlásfüggvényt:

(%i26)
```
wxplot2d([y1num,y2num],[x,0,L],[legend,false]),adat;
```

(%t26) 

(%o26)

Látható, hogy a B helyen az értékük és deriváltjaiknak értéke is azonos. Ábrázoljuk a függvényeket csak a rájuk vonatkozó tartományokon:

(%i27)
```
ykozos: if x < L/2 then y1num else y2num, adat$
```

(%i28)
```
wxplot2d(ykozos,[x,0,L]),adat;
```

(%t28) 

(%o28)

Nézzük meg hasonlóképpen a szögelfordulás függvényt:

(%i30)
```
fikozos: if x < L/2 then fi1num else fi2num, adat$
wxplot2d(fikozos,[x,0,L]),adat;
```

(%t30)



(%o30)

Külön ablakban megjelenítve őket:

(%i31)
```
plot2d(
    ykozos,[x,0,L],
    [xlabel,"x [m]"],
    [ylabel, "y [m]"],
    [title, "A lehajlásfüggvény"]
),adat;
```

(%o31) [*c:/temp/maxout7272.gnuplot*]

→
```
plot2d(
    fikozos,[x,0,L],
    [xlabel,"x [m]"],
    [ylabel, "fi [rad]"],
    [title, "A szögelfordulás"]
),adat;
```

(%o58) [*c:/temp/maxout8328.gnuplot*]

4 Szélsőérték számítása az AB szakaszon

Látható, hogy az AB szakaszon szélsőértéke van egy ismeretlen x0 helyen a lehajlásfüggvénynek. Ezen a helyen a szögelfordulás zérus. Tehát x0 könnyen számítható:

(%i55)
```
fi1par:fi1,cmego$
x0sol:solve(fi1par=0,x);
```

(x0sol) $[x=-\frac{L}{2\sqrt{3}}, x=\frac{L}{2\sqrt{3}}]$

A kapott megoldásokból a pozitív érték adja x0 értékét. Ezt akár numerikusan is számíthatjuk gyökkereséssel:

(%i49)
```
x0:find_root(fi1num,0,L/2),adat;
```

(x0) 0.8660254037844386

x0 helyen a lehajlás értéke:

(%i65)
```
y1,cmego,x0sol[2],fullratsimp;
%,adat,float;
```

(%o64) $\frac{F L^3}{8\, 3^{5/2}\, IE}$

(%o65) 0.00757772228311384

vagy:

(%i67)
```
y1num,x=x0;
```

(%o67) 0.007577722283113838